;   Example script to produce plots for a WRF real-data run,
;   with the ARW coordinate dynamics option.


load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl"
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"
load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/contributed.ncl"
load "$NCARG_ROOT/lib/ncarg/nclscripts/wrf/WRFUserARW.ncl"

;load "./WRFUserARW.ncl"

begin
;
; Make a list of all files we are interested in
  DATADir = "/scratch/02126/hargrove/nests/member_52/"
  FILES = systemfunc (" ls -1 " + DATADir + "wrfout_d03_* ")
  numFILES = dimsizes(FILES)
  print("numFILES = " + numFILES)
  print(FILES)
  print (" ")

; We generate plots, but what kind do we prefer?
; type = "x11"
; type = "pdf"
  type = "ps"
; type = "ncgm"
  wks = gsn_open_wks(type,"precip")


; Set some basic resources
  res = True
  res@MainTitle = "REAL-TIME WRF"

  pltres = True
  mpres = True
  mpres@mpGeophysicalLineColor = "Black"
  mpres@mpNationalLineColor    = "Black"
  mpres@mpUSStateLineColor     = "Black"
  mpres@mpGridLineColor        = "Black"
  mpres@mpLimbLineColor        = "Black"
  mpres@mpPerimLineColor       = "Black"

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
  f = addfiles(FILES+".nc","r")

  times = wrf_user_getvar(f,"times",-1)  ; get all times in the file
ntimes = dimsizes(times)         ; number of times in the file

  ; Get non-convective, convective and total precipitation
   rain_exp = wrf_user_getvar(f,"RAINNC",-1)
   rain_con = wrf_user_getvar(f,"RAINC",-1)
   rain_tot = rain_exp + rain_con
   rain_tot = rain_tot*0.0394
   rain_tot@description = "Total Precipitation"


;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

  do it = 1,numFILES-1

    print("Working on time: " + times(it) )
    res@TimeLabel = times(it)   ; Set Valid time to use on plots

  ; Plotting options for Precipitation
    opts_r = res
    opts_r@UnitLabel            = "in"
    opts_r@cnLevelSelectionMode = "ExplicitLevels"
    opts_r@cnLevels             = (/ .1, .25, .5, .75, 1.,1.25, 1.5, \
                                    2., 3., 4., 5./)
    opts_r@cnFillColors         = (/"White","White","DarkOliveGreen1", \
                                "DarkOliveGreen3","Chartreuse", \
                                    "Chartreuse3","Green","ForestGreen", \
                                    "Yellow","Orange","Red","Violet"/)
    opts_r@cnInfoLabelOn        = False
    opts_r@cnConstFLabelOn      = False
    opts_r@cnFillOn             = True


  ; Total Precipitation (color fill)
    contour_tot = wrf_contour(f[it],wks, rain_tot(it,:,:), opts_r)

    delete(opts_r)



  ; MAKE PLOTS

    ; Total Precipitation
      plot = wrf_map_overlays(f[it],wks,contour_tot,pltres,mpres)


;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

  end do        ; END OF TIME LOOP

end

